NASA TECHNICAL REPORT

NASA TR R-458

NASA TR R-458

AN EXACT TRANSFORMATION FROM GEOCENTRIC TO GEODETIC COORDINATES FOR NONZERO ALTITUDES

David R. Hedgley, Jr.

Flight Research Center
Edwards, Calif. 93523

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • MARCH 1976



1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
TR R-458		

4. Title and Subtitle	5. Report Date
AN EXACT TRANSFORMATION FROM GEOCENTRIC TO GEODETIC COORDINATES FOR NONZERO ALTITUDES	March 1976
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
David R. Hedgley, Jr.	H-909
	10. Work Unit No.
9. Performing Organization Name and Address	312-04-02
NASA Flight Research Center P. O. Box 273 Edwards, California 93523	11. Contract or Grant No.
	13. Type of Report and Period Covered
12. Sponsoring Agency Name and Address	Technical Report
National Aeronautics and Space Administration Washington, D. C. 20546	14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

An exact method for the nonzero altitude transformation from geocentric to geodetic coordinates is derived. The method is mathematically general and should serve as a primary standard.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Geodetic Geocentric	Unclassified - Unlimited Category: 01

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified	18	$3.25

* For sale by the National Technical Information Service, Springfield, Virginia 22161



AN EXACT TRANSFORMATION FROM GEOCENTRIC TO GEODETIC COORDINATES FOR NONZERO ALTITUDES

David R. Hedgley, Jr.
Flight Research Center

INTRODUCTION

In space position measurement systems, it is often necessary to transform geocentric to geodetic coordinates. Consequently, it is important to have an exact closed-form solution for the transformation that is free of embedded singularities. An exact solution is particularly important for determining missile positions. All the previous work in this area falls into two categories: formulas that are mathematically exact but involve equations that become unstable in the neighborhood of their singularities, and formulas that are approximations that may or may not have inherent singularities. References 1 and 2 are of the former type, and references 3 to 5 are of the latter.

Although some of these methods are adequate under some conditions, they have limitations under certain circumstances. This paper presents an exact closed-form solution for the transformation that is free of singularities. The solution permits the evaluation of any of the existing methods under any conditions and should serve as a primary standard.

The appendix contains computer subroutines that implement the ideas that follow and examples of input and output data.

SYMBOLS

Parenthetical symbols are computer identifiers for variables.

a (A)	major axis of ellipsoid
b (B)	minor axis of ellipsoid
d	the distance from any point in space to the ellipsoid
d_x	$= x_0 - x_2$

d_y	$= y_0 - y_2$
d_z	$= z_0 - z_2$
e (E)	eccentricity
f (F)	flattening factor
$f(x,y,z)$	general function to be minimized
g	restraint imposed on f
H	function which is composite of functions f and g
h (ALT)	altitude normal to the ellipsoid
h_s (ALTS)	altitude computed from distance formula
p_0	any point in space
p_2	the point on the ellipsoid that is the minimum distance from the point p_0
t_x	percentage of error in d_x
t_y	percentage of error in d_y
t_z	percentage of error in d_z
X	coordinate axis that intercepts the Greenwich meridian
x,y,z	coordinates of any point on the ellipsoid
x_0,y_0,z_0 (X0,Y0,Z0)	coordinates of p_0
x_2,y_2,z_2 (X2,Y2,Z2)	coordinates of p_2
Y	coordinate axis that is in a direction normal to the plane determined by the intersection of X and Z
Z	coordinate axis that intercepts the poles

α (V)	arbitrary real constant to be determined
ε	error in α
λ_e (ALONG)	longitude east
λ_w (ALONG)	longitude west
φ (ALAT)	geodetic latitude
ω	geocentric latitude

ANALYTICAL FORMULATION

Let a model of the earth be an ellipsoid given by the following equation (ref. 6):

$$\frac{x^2}{a^2} + \frac{y^2}{a^2} + \frac{z^2}{b^2} = 1 \tag{1}$$

where $a > b$. In addition, let $p_0\ (x_0, y_0, z_0)$ be any point above the ellipsoid, as shown in the sketch below.



The line perpendicular to the surface of the ellipsoid from p_0 is the altitude of p_0 above the ellipsoid and hence is the shortest distance from p_0 to that surface (eq. (1)).

Therefore, by minimizing the distance from p_0 to the surface of the ellipsoid, it is possible to acquire the coordinates (x_2, y_2, and z_2) of the point on the surface that cause the distance defined by

$$d = \left[(x - x_0)^2 + (y - y_0)^2 + (z - z_0)^2\right]^{1/2} \tag{2}$$

to be minimum.

The Lagrange multiplier method (ref. 7) is used to implement this minimization where

$$f(x,y,z) = (x - x_0)^2 + (y - y_0)^2 + (z - z_0)^2$$

and

$$g(x,y,z) = \frac{x^2}{a^2} + \frac{y^2}{a^2} + \frac{z^2}{b^2} - 1$$

Then

$$H(x,y,z,\alpha) = (x - x_0)^2 + (y - y_0)^2 + (z - z_0)^2 - \alpha\left(\frac{x^2}{a^2} + \frac{y^2}{a^2} + \frac{z^2}{b^2} - 1\right)$$

If partial derivatives are taken in turn and each is equated to zero, the following equations result:

$$\frac{\partial H}{\partial x} = 2(x - x_0) - \frac{\alpha 2x}{a^2} = 0 \text{ or } x = x_0/(1 - \alpha/a^2) \tag{3}$$

$$\frac{\partial H}{\partial y} = 2(y - y_0) - \frac{\alpha 2y}{a^2} = 0 \text{ or } y = y_0/(1 - \alpha/a^2) \tag{4}$$

$$\frac{\partial H}{\partial z} = 2(z - z_0) - \frac{\alpha 2z}{b^2} = 0 \text{ or } z = z_0/(1 - \alpha/b^2) \tag{5}$$

$$\frac{\partial H}{\partial \alpha} = -\frac{x^2}{a^2} - \frac{y^2}{a^2} - \frac{z^2}{b^2} + 1 = 0 \tag{6}$$

Equations (3) to (5) can be substituted into equation (6) as follows:

$$-\left[x_0/\left(1-\alpha/a^2\right)\right]^2/a^2-\left[y_0/\left(1-\alpha/a^2\right)\right]^2/a^2-\left[z_0/\left(1-\alpha/b^2\right)\right]^2/b^2+1=0$$

or

$$-x_0{}^2\left(b^2-2\alpha+\alpha^2/b^2\right)-y_0{}^2\left(b^2-2\alpha+\alpha^2/b^2\right)-z_0{}^2\left(a^2-2\alpha+\alpha^2/a^2\right)+\left(a^2-2\alpha-\alpha^2/a^2\right)\left(b^2-2\alpha+\alpha^2/b^2\right)=0$$

Simplifying and collecting,

$$\begin{aligned}&\left(1/a^2b^2\right)\alpha^4-2\left(1/a^2+1/b^2\right)\alpha^3+\left(4+a^2/b^2+x_0{}^2/b^2-y_0{}^2/b^2-z_0{}^2/a^2\right)\alpha^2\\&+2\left(x_0{}^2+y_0{}^2+z_0{}^2-a^2-b^2\right)\alpha+a^2b^2-x_0{}^2b^2-y_0{}^2b^2-z_0{}^2a^2=0\end{aligned}\tag{7}$$

This is a quartic equation in α and can be solved in closed form (ref. 8). The proper solution can be determined if the solution is restricted to the real zeros of equation (7) and that zero of equation (7) is chosen that causes d to be smallest. The appropriate solution of this quartic has no singularities. That is, α does not approach plus or minus infinity, because that would imply from equations (3) to (5) that x_2, y_2, and z_2 equal zero. Equations (3) to (5) do not become unstable, because the statement $\alpha = a^2$ or $\alpha = b^2$ implies that x_2, y_2, or z_2 approaches infinity. Since altitude is nonzero, α does not equal zero.

Let α_0 be that zero that satisfies the quartic equation. Then, from equations (3) to (5),

$$x_2=x_0/\left(1-\alpha_0/a^2\right)=x_0a^2/\left(a^2-\alpha_0\right)\tag{8}$$

$$y_2=y_0/\left(1-\alpha_0/a^2\right)=y_0a^2/\left(a^2-\alpha_0\right)\tag{9}$$

$$z_2=z_0/\left(1-\alpha_0/b^2\right)=z_0b^2/\left(b^2-\alpha_0\right)\tag{10}$$

$$h_s=\left[\left(x_2-x_0\right)^2+\left(y_2-y_0\right)^2+\left(z_2-z_0\right)^2\right]^{1/2}\tag{11}$$

$$\varphi=\sin^{-1}\left(z_0-z_2\right)/h_s\tag{12}$$

$$\lambda_e=\tan^{-1}\left(y_2/x_2\right)=\tan^{-1}\left(y_0/x_0\right)=-\lambda_w\tag{13}$$

Although the foregoing formulas are mathematically exact, for computational reasons it is usually advantageous to normalize all distances, because this minimizes errors due to rounding off. However, errors due to rounding off are still possible. To circumvent this problem for altitudes not near zero, consider the following error analysis with respect to the computation of h_s, λ, and φ.

Let ε be a small error introduced in the computation of the zero of the quartic. In addition, let $d_z = z_0 - z_2 = -z_0\alpha_0/\left(b^2 - \alpha_0\right)$. Then the error in d_z is introduced by α_0 only, because a, b, x_0, y_0, and z_0 are all exact. It is easy to show that the percentage of error, t_z, of d_z is as follows:

$$t_z = -1 + \left(1 + \varepsilon/\alpha_0\right)\left[\left(b^2 - \alpha_0\right)/\left(b^2 - \alpha_0 - \varepsilon\right)\right]$$

From equation (10), $b^2 - \alpha_0 > b^2$. Therefore $\left|b^2 - \alpha_0\right| > \left|\alpha_0\right| > \left|\varepsilon\right|$, and since ε is small, $\left(b^2 - \alpha_0\right)/\left(b^2 - \alpha_0 - \varepsilon\right) \approx 1$. Therefore $t_z \approx \varepsilon/\alpha_0$. Similarly, $t_x, t_y \approx \varepsilon/\alpha_0$.

Since equation (12) involves the ratio of two numbers with virtually equal percentages of error, latitude is undisturbed by a small error in the computation of α_0 due to rounding off.

As noted above, this analysis is valid only if the altitude does not approach zero; otherwise α_0 approaches zero and the value $\left|\varepsilon/\alpha_0\right|$ may approach infinity. The instability of $\left|\varepsilon/\alpha_0\right|$ suggests that equation (12) may become unstable under these conditions.

This final problem can be resolved. If $p_2\ (x_2, y_2, z_2)$ is the point on the ellipsoid that intersects the normal drawn from $p_0\ (x_0, y_0, z_0)$, the following equation (ref. 6) may be used:

$$\varphi = \tan^{-1}\left[\tan\omega/\left(1 - e^2\right)\right] \tag{14}$$

where $\tan\omega = z_2/\left(x_2^2 + y_2^2\right)^{1/2}$. From these relationships, φ may be determined immediately. The error, if any, in $\tan\omega$ introduced by the computational round-off error in α_0 is inconsequential, because it can easily be shown that the percentage of error in x_2, y_2, and z_2 is virtually zero. In fact, the percentage of error in z_2 and x_2 or y_2 is given by $\left(b^2 - \alpha_0\right)/\left(b^2 - \alpha_0 - \varepsilon\right) - 1$ and $\left(a^2 - \alpha_0\right)/\left(a^2 - \alpha_0 - \varepsilon\right) - 1$, respectively. Since ε is small for any value of α_0, the quantities

$\left(b^2 - \alpha_0\right)/\left(b^2 - \alpha_0 - \varepsilon\right)$ and $\left(a^2 - \alpha_0\right)/\left(a^2 - \alpha_0 - \varepsilon\right)$ approximate 1 under any conditions. Thus, equation (14) is preferable to equation (12) for determining geodetic latitude for any altitude.

However, even the smallest error in α_0 causes more than negligible error in the computation of h_s (eq. (11)), because the error is amplified by the scaling factor used to restore the true value of h after computations are completed. This analysis suggests that it is preferable to use the following exact formulas from reference 6 to compute h instead of equation (11):

$$x_0 = \left[a/\left(1 - e^2 \sin^2 \varphi\right)^{1/2} + h\right] \cos \varphi \cos \lambda$$

$$y_0 = \left[a/\left(1 - e^2 \sin^2 \varphi\right)^{1/2} + h\right] \cos \varphi \sin \lambda$$

$$z_0 = \left[a\left(1 - e^2\right)/\left(1 - e^2 \sin^2 \varphi\right)^{1/2} + h\right] \sin \varphi$$

where $f = (a - b)/a$ and $e^2 = 2f - f^2$. Since latitude and longitude are known, h can be computed directly and the proper equation can be chosen to compute h to avoid division by values close to zero.

In this way the geodetic coordinates φ, λ_w, and h are determined exactly both mathematically and computationally. Therefore, the computational accuracy of the method is the accuracy of the computer used.

CONCLUDING REMARKS

A mathematically exact method for computing geodetic coordinates from geocentric coordinates is derived. The computational accuracy achieved by using the method is as accurate as the computer used. The transformation provides a primary standard and makes possible the evaluation of any of the existing methods.

Flight Research Center
National Aeronautics and Space Administration
Edwards, Calif., November 4, 1975

APPENDIX

COMPUTER SUBROUTINES AND DATA SAMPLES

The following computer subroutines implement the theory presented in the text. The subroutines were written in FORTRAN IV.

The comments in the listings should be sufficient for their comprehension and modification. A sample input and output listing is provided to facilitate the verification of the correct FORTRAN code.



Subroutine GEOD

```
      SUBROUTINE GEOD(ELE,ELEM,ELES,AZI,AZIM,AZIS,PHI,PHIM,PHIS,
     *ALM,ALMM,ALMS,RAN,H,ALAT,ALONG,ALT)
C
C
C     THIS SUBROUTINE COMPUTES THE ALTITUDE, GEODETIC LATITUDE AND
C     LONGITUDE GIVEN THE RANGE, AZIMUTH AND ELEVATION OF A POINT
C     WITH RESPECT TO A RADAR SITE.
C
C
C     ELE,ELEM,ELES ARE THE DEGREES, MINUTES AND SECONDS RESPECTIVELY OF
C     ELEVATION OF THE TARGET WITH RESPECT TO A RADAR SITE.
C
C     AZI,AZIM,AZIS ARE THE DEGREES, MINUTES AND SECONDS RESPECTIVELY OF
C     AZIMUTH OF THE TARGET WITH RESPECT TO A RADAR SITE.
C
C     PHI,PHIM,PHIS ARE THE DEGREES, MINUTES AND SECONDS RESPECTIVELY
C     OF THE GEODETIC LATITUDE OF THE RADAR SITE.
C
C     ALM,ALMM,ALMS ARE THE DEGREES, MINUTES AND SECONDS RESPECTIVELY
C     OF THE LONGITUDE OF THE RADAR SITE.
C
C     RAN IS THE RANGE OF THE TARGET WITH RESPECT TO THE RADAR SITE.
C     H IS THE ALTITUDE OF THE RADAR SITE ABOVE THE REFERENCE ELLIPSOID.
C
C     ALAT IS THE COMPUTED GEODETIC LATITUDE.
C     ALONG IS THE COMPUTED LONGITUDE.
C     ALT IS THE COMPUTED ALTITUDE.
C
      DIMENSION R(12),ROOT(12)
C
C     DEFINE MAJOR AND MINOR AXIS OF ELLIPSOID IN FEET.
C
      A=20925832.
      B=20854892.
      F=(A-B)/A
      CE=2*F-F**2
C
C     NORMALIZE IN UNITS OF A.
C
      ZP=A
      A1=A
      H=H/ZP
      RAN=RAN/ZP
      A=A/ZP
      B=B/ZP
      GK=3.1415926536
      CON=GK/180.
C
C        CONVERT ANGLES FROM DEGREES TO RADIANS.
C
      ELE=ELE+(ELEM/60)+(ELES/3600)
      PHI=PHI+(PHIM/60)+(PHIS/3600)
      AZI=AZI+(AZIM/60)+(AZIS/3600)
      ALM=ALM+(ALMM/60)+(ALMS/3600)
      ALM=360-ALM
      PHI=PHI*CON
      ALM=ALM*CON
```

```
      AZI=AZI*CON                                                       58
      ELE=ELE*CON                                                       59
60    C                                                                 60
      C                                                                 61
      C     COMPUTE THE TRANSFORMATION OF TARGET COORDINATES,           62
      C     FROM POLAR COORDINATES TO CARTESIAN COORDINATES,            63
      C     CENTERED AT TRACKING RADAR.                                 64
65    C                                                                 65
      XO=RAN*SIN(AZI)*COS(ELE)                                          66
      ZO=RAN*SIN(ELE)                                                   67
      YO=RAN*COS(AZI)*COS(ELE)                                          68
      C                                                                 69
70    C     COMPUTE THE TRANSFORMATION OF THE RADAR SITE GIVEN          70
      C     IN GEODETIC POLAR COORDINATES TO CARTESIAN COORDINATES      71
      C     WITH RESPECT TO THE GEOCENTRIC CENTER OF THE EARTH.         72
      C                                                                 73
      EX=1/(1-CE*(SIN(PHI)**2))                                         74
75    EX=ABS(EX)                                                        75
      EX=EX**.5                                                         76
      EX=EX*A                                                           77
      EO=(EX+H)*COS(PHI)*COS(ALM)                                       78
      FO=(EX+H)*COS(PHI)*SIN(ALM)                                       79
80    GO=(EX*(1-CE)+H)*SIN(PHI)                                         80
      C                                                                 81
      C     COMPUTES THE DIRECTION COSINES OF THE AXIS OF THE RADAR SITE. 82
      C                                                                 83
      U1=-SIN(ALM)                                                      84
85    U2=COS(ALM)                                                       85
      U3=0                                                              86
      V1=-COS(ALM)*SIN(PHI)                                             87
      V2=-SIN(ALM)*SIN(PHI)                                             88
      V3=COS(PHI)                                                       89
90    W1=COS(ALM)*COS(PHI)                                              90
      W2=SIN(ALM)*COS(PHI)                                              91
      W3=SIN(PHI)                                                       92
      C                                                                 93
      C     COMPUTE THE TRANSFORMATION OF THE POINT OR TARGET           94
95    C     FROM CARTESIAN COORDINATES CENTERED AT TRACKING             95
      C     RADAR TO GEOCENTRIC CARTESIAN COORDINATES.                  96
      C                                                                 97
      X5=XO*U1+YO*V1+ZO*W1+EO                                           98
      Y5=XO*U2+YO*V2+ZO*W2+FO                                           99
100   Z5=XO*U3+YO*V3+ZO*W3+GO                                           100
      XO=X5                                                             101
      YO=Y5                                                             102
      ZO=Z5                                                             103
      C                                                                 104
105   C     CONSTRUCT THE COEFFICIENTS OF THE QUARTIC EQUATION:         105
      C           R(5) * X**4 + R(4) * X**3 + R(3) * X**2 +             106
      C           R(2) * X + R(1) = 0                                   107
      C                                                                 108
      R(5)=1/((A*B)**2)                                                 109
110   R(4)=-2*((1/(A**2))+(1/(B**2)))                                   110
      R(3)=4+((B/A)**2)+((A/B)**2)-((XO/B)**2)-((YO/B)**2)-((ZO/A)**2)  111
      R(2)=2*(XO**2+YO**2+ZO**2)-2*(A**2+B**2)                          112
      R(1)=(A*B)**2-(XO*B)**2-(YO*B)**2-(ZO*A)**2                       113
      C                                                                 114
```

```
C        CALL QUART TO SOLVE THE QUARTIC EQUATION FOR ALL
C        REAL ZEROS WHERE:
C              R = ARRAY OF COEFFICIENTS
C           ROOT = ARRAY OF REAL ZEROS
C             NI = NUMBER OF REAL ZEROS
C
      CALL QUART(R,ROOT,NI)
C
C     DETERMINE THE COORDINATES ON THE ELLIPSOID THAT
C     MAKES THE DISTANCE FROM THE TARGET TO THE SURFACE A MINIMUM.
C
      PS=10.**30
      XJ = 0.0
      YJ = 0.0
      ZJ = 0.0
      DO 16 J=1,NI
      V=ROOT(J)
      X2=(XO*A**2)/(A**2-V)
      Y2=(YO*A**2)/(A**2-V)
      Z2=(ZO*B**2)/(B**2-V)
      XA=(-XO*V)/(A**2-V)
      YA=(-YO*V)/(A**2-V)
      ZA=(-ZO*V)/(B**2-V)
      U=XA**2+YA**2+ZA**2
      U=U**.5
      IF(U.GT.PS) GO TO 16
      PS=U
      XJ=X2
      YJ=Y2
      ZJ=Z2
   16 CONTINUE
      X2=XJ
      Y2=YJ
      Z2=ZJ
C
C        THE ALTITUDE EQUALS THE MINIMUM DISTANCE FROM THE
C        TARGET TO THE SURFACE.
      CON=180./GK
C
C     DETERMINE THE LONGITUDE.
C
      ALONG=ATAN2(YO,XO)
C
C     DETERMINE THE GEODETIC LATITUDE.
C
      D=X2**2+Y2**2
      D=D**.5
      AA=D*(1.-CE)
      ALAT=ATAN2(Z2,AA)
      ALAT=SIGN(ALAT,ZO)
      X1=XO*ZP
      Y1=YO*ZP
      Z1=ZO*ZP
      EX=1./(1.-CE*(SIN(ALAT)**2))
      EX=ABS(EX)
      EX=EX**.5
      EX=EX*A1
```

```
      C                                                                 172
      C     COMPUTE ALTITUDE BY AN EXACT EQUATION WHICH MINIMIZES ROUND OFF 173
      C     ERRORS.                                                     174
175   C                                                                 175
            IF(ABS(SIN(ALAT)).GT..1)GO TO 14                            176
            IF(ABS(COS(ALONG)).GT..1)GO TO 15                           177
            ALT=(Y1/(COS(ALAT)*SIN(ALONG)))-EX                          178
            GO TO 710                                                   179
180        14 ALT=(Z1/SIN(ALAT))-(EX*(1-CE))                            180
            GO TO 710                                                   181
           15 ALT=(X1/(COS(ALAT)*COS(ALONG)))-EX                        182
          710 CONTINUE                                                  183
            ALAT=ALAT*CON                                               184
185   C                                                                 185
      C     CHANGE SIGN OF LONGITUDE TO CONFORM WITH INPUT WHICH        186
      C     WAS IN DEGREES WEST...CALCULATIONS GIVE LONGITUDE IN        187
      C     DEGREES EAST, CHANGE SIGN TO GET LONGITUDE IN DEGREES       188
      C     WEST, CONVERT FROM RADIANS TO DEGREES.                      189
190   C                                                                 190
            ALONG=-ALONG*CON                                            191
      C                                                                 192
            RETURN                                                      193
            END                                                         194
```

Subroutine QUART

```
      SUBROUTINE QUART(R,ROOT,NI)                                          195
C                                                                          196
C     THIS ROUTINE SOLVES FOR THE REAL ZEROS ONLY VIA                      197
C     FERRARI'S METHOD. (NEW FIRST COURSE IN THE THEORY OF                 198
C     EQUATIONS, DICKSON, PP. 51-52)                                       199
C                                                                          200
C                                                                          201
C                                                                          202
C     R IS THE ARRAY WHICH CONTAINS THE COEFFICIENTS OF THE                203
C     QUARTIC ARRANGED IN ASCENDING ORDER.                                 204
C             R(5) * X**4 + R(4) * X**3 +                                  205
C             R(3) * X**2 + R(2) * X + R(1) = 0                            206
C                                                                          207
C     THE ARRAY ROOT WILL CONTAIN THE REAL ZEROS OF THE QUARTIC            208
C                                                                          209
C     NI IS THE SCALAR VARIABLE THAT STATES THE NUMBER OF REAL ZEROS       210
C                                                                          211
C                                                                          212
C                                                                          213
      DIMENSION R(1),ROOT(1)                                               214
      NI=0                                                                 215
C                                                                          216
C        NORMALIZE COEFFICIENTS SO:                                        217
C           X**4 + B*X**3 + C*X**2 + D*X + E = 0                           218
C                                                                          219
      DO 19 J=1,5                                                          220
   19 R(J)=R(J)/R(5)                                                       221
      B=R(4)                                                               222
      C=R(3)                                                               223
      D=R(2)                                                               224
      E=R(1)                                                               225
C                                                                          226
C        CALCULATE COEFFICIENTS B1, C1, D1, OF THE RESOLVENT CUBIC         227
C           Y**3 + B1*Y**2 + C1*Y + D1 = 0                                 228
      B1=-C                                                                229
      C1=B*D-4*E                                                           230
      D1=-(B**2)*E+4*C*E-D**2                                              231
C                                                                          232
C        IN SOLVING CUBIC EQUATION, CALCULATE COEFFICIENTS                 233
C        OF THE CORRESPONDING REDUCED CUBIC WHICH HAS NO TERM              234
C        OF THE SECOND DEGREE BY SETTING Y = Z - B1/3:                     235
C           Z**3 + P*Z + Q = 0                                             236
C                                                                          237
      P=C1-((B1**2)/3)                                                     238
      Q=D1-((B1*C1)/3)+((2*(B1**3))/27)                                    239
C                                                                          240
C        THE DISCRIMINANT OF THE GENERAL CUBIC EQUATION IS                 241
C        EQUAL TO THE DISCRIMINANT DEL OF THE CORRESPONDING                242
C        REDUCED EQUATION.                                                 243
C                                                                          244
      DELT=18*B1*C1*D1-4*(B1**3)*D1+(B1*C1)**2                             245
      DELV=-4*(C1**3)-27*(D1**2)                                           246
      DEL=DELT+DELV                                                        247
C                                                                          248
C        IF DEL IS NEGATIVE, ONE ROOT IS REAL, AND TWO ARE                 249
C        CONJUGATE IMAGINARIES.                                            250
C                                                                          251
```

```
      IF(DEL.GE.0.)GO TO 12                                              252
      RS=(P/3)**3+(Q/2)**2                                               253
60    RS=ABS(RS)                                                         254
      A=((-Q/2)+RS**.5)                                                  255
      W=((-Q/2)-(RS**.5))                                                256
      S=ABS(A)                                                           257
      T=ABS(W)                                                           258
65    W1=W                                                               259
      A1=A                                                               260
      VV=1./3.                                                           261
      A=S**VV                                                            262
      W=T**VV                                                            263
70    A=A*SIGN(1.,A1)                                                    264
      W=W*SIGN(1.,W1)                                                    265
C                                                                        266
C        Y1 IS THE SINGLE REAL ROOT OF THE REDUCED CUBIC,                267
C        Y IS THE SINGLE REAL ROOT OF THE GENERAL CUBIC.                 268
75    C                                                                  269
      Y1=A+W                                                             270
      Y=Y1-(B1/3)                                                        271
      GO TO 10                                                           272
   12 CONTINUE                                                           273
80    C                                                                  274
C        IF DEL IS POSITIVE, THERE ARE THREE DISTINCT REAL               275
C        ROOTS. IF DEL IS ZERO, THERE ARE AT LEAST TWO EQUAL             276
C        REAL ROOTS.                                                     277
C             TRIGONOMETRIC SOLUTION IS USED.                            278
85    C                                                                  279
      TN=(-4*P/3)                                                        280
      TN=ABS(TN)                                                         281
      TN=TN**.5                                                          282
       COS3A=-.5*Q*((-3/P)**1.5)                                         283
90    IF(ABS(COS3A).GT.1.) COS3A=SIGN(1.,COS3A)                          284
      ARC=ACOS(COS3A)/3                                                  285
      COSA=COS(ARC)                                                      286
      Y1=TN*COSA                                                         287
      Y=Y1-(B1/3)                                                        288
95 10 CONTINUE                                                           289
C                                                                        290
C        BACK TO SOLVING THE QUARTIC, WHERE Y IS SUCH THAT               291
C           A2*X**2 + B2*X + C2    IS THE SQUARE OF A LINEAR FUNCTION,   292
C        M*X + N, AND ALSO EQUAL TO  (X**2 + B*X/2 + Y/2)**2             293
100   C                                                                  294
      A2=.25*(B**2)-C+Y                                                  295
      B2=.5*B*Y-D                                                        296
      C2=.25*(Y**2)-E                                                    297
C                                                                        298
105   IF(A2.NE.0.)GO TO 111                                              299
C                                                                        300
C        IN CASE A2=0,  (X**2 + B*X/2 + Y/2)**2 = B2*X + C2              301
C        SINCE THE POLYNOMIAL IS A PERFECT SQUARE, B2 = 0                302
C        AND THE QUADRATICS TO BE SOLVED ARE:                            303
110   C           X**2 + B*X/2 + Y/2 - C2**.5 = 0    AND                  304
C           X**2 + B*X/2 + Y/2 + C2**.5                                  305
C                                                                        306
      DEL1 = (B/2)**2 - 4*(Y/2 - C2**.5)                                 307
      DEL2 = (B/2)**2 - 4*(Y/2 + C2**.5)                                 308
```

```
115                DX1=B/2                                                    309
                   DX2=B/2                                                    310
                   GO TO 223                                                  311
                 111 CONTINUE                                                 312
               C                                                              313
120            C        IN CASE  A2  IS NOT 0, THEN   M = A2**.5,  N = B2/(2*M),  314
               C        THEN THE QUADRATICS TO BE SOLVED ARE:                 315
               C           X**2 + (B/2 - M)*X + Y/2 - N = 0    AND            316
               C           X**2 + (B/2 + M)*X + Y/2 + N = 0                   317
               C                                                              318
125                A2=ABS(A2)                                                 319
                   AM=A2**.5                                                  320
                   AN=B2/(2*AM)                                               321
                   DX1=.5*B-AM                                                322
                   DX2=.5*B+AM                                                323
130                DEL1 = (B/2 - AM)**2 - 4*(Y/2 - AN)                        324
                   DEL2 = (B/2 + AM)**2 - 4*(Y/2 + AN)                        325
                 223 CONTINUE                                                 326
               C                                                              327
               C        ROOTS OF TWO QUADRATICS ARE THE FOUR ROOTS OF THE QUARTIC,  328
135            C             ONLY THE REAL ROOTS ARE RETURNED.                329
               C                                                              330
                   IF(DEL1.LT.0.)GO TO 22                                     331
                   NI=NI+1                                                    332
                   ROOT(NI)=(-DX1/2)+((DEL1**.5)/2)                           333
140                NI=NI+1                                                    334
                   ROOT(NI)=(-DX1/2)-((DEL1**.5)/2)                           335
                  22 IF(DEL2.LT.0.)GO TO 25                                   336
                   NI=NI+1                                                    337
                   ROOT(NI)=(-DX2/2)+((DEL2**.5)/2)                           338
145                NI=NI+1                                                    339
                   ROOT(NI)=(-DX2/2)-((DEL2**.5)/2)                           340
                  25 CONTINUE                                                 341
                   RETURN                                                     342
                   END                                                        343
```

Input and Output Data Samples

```
INPUT-----
          POINT COORDINATES WRT RADAR SITE
                    ELEVATION DEG.                    90.0000
                    AZI       DEG.                    -0.0000
                    RANGE     FEET                      .1000

          RADAR SITE COORDINATES
                    GEODETIC LATITUDE                 55.0000
                    LONGITUDE                         22.0000
                    ALTITUDE      FEET                -0.0000

OUTPUT----
          POINT COORDINATES
                    GEODETIC LATITUDE                 55.0000      .550000000006E+02
                    LONGITUDE                         22.0000      .219999999988E+02
                    ALTITUDE      FEET                  .0999      .998557806015E-01



INPUT-----
          POINT COORDINATES WRT RADAR SITE
                    ELEVATION DEG.                    90.0000
                    AZI       DEG.                    -0.0000
                    RANGE     FEET                 50000.0000

          RADAR SITE COORDINATES
                    GEODETIC LATITUDE                 44.0000
                    LONGITUDE                        121.0000
                    ALTITUDE      FEET               500.0000

OUTPUT----
          POINT COORDINATES
                    GEODETIC LATITUDE                 44.0000      .440000000264E+02
                    LONGITUDE                        121.0000      .120999999999E+03
                    ALTITUDE      FEET             50499.9900      .504999899952E+05



INPUT-----
          POINT COORDINATES WRT RADAR SITE
                    ELEVATION DEG.                    90.0000
                    AZI       DEG.                    -0.0000
                    RANGE     FEET              999999999.0000

          RADAR SITE COORDINATES
                    GEODETIC LATITUDE                 33.0000
                    LONGITUDE                         88.0000
                    ALTITUDE      FEET               700.0000

OUTPUT----
          POINT COORDINATES
                    GEODETIC LATITUDE                 33.0000      .329999999979E+02
                    LONGITUDE                         88.0000      .879999999988E+02
                    ALTITUDE      FEET          100000699.0492      .100000069905E+10
```

REFERENCES

1. Paul, M. K.: A Note on Computation of Geodetic Coordinates From Geocentric (Cartesian) Coordinates. Bulletin Geodesique, nouvelle série, no. 108, June 1, 1973, pp. 135-139.

2. Sugai, I.: Exact Geodetic Latitude of Subvehicle Point. J. Astronaut. Sci., vol. XXII, no. 1, July-Sept. 1974, pp. 55-63.

3. Long, Sheila Ann T.: Derivation of Transformation Formulas Between Geocentric and Geodetic Coordinates for Nonzero Altitudes. NASA TN D-7522, 1974.

4. Taylor, William C.; Everett, C. R.; and Bryant, Ermon H.: Geodetic and Orthogonal Coordinate Transformations. APGC Tech. Paper No. 7, Air Proving Ground Center, Eglin Air Force Base, Jan. 1960, pp. 11-12.

5. Morrison, John; and Pines, Samuel: The Reduction From Geocentric to Geodetic Coordinates. Astron. J., vol. 66, no. 1, Feb. 1961, pp. 15-16.

6. Escobal, Pedro Ramon: Methods of Orbit Determination. John Wiley & Sons, Inc., c.1965, p. 23.

7. Thomas, George B., Jr.: Calculus and Analytic Geometry. Part two. Fourth ed., Addison-Wesley Publishing Co., Dec. 1969, pp. 528-530.

8. Dickson, Leonard Eugene: New First Course in the Theory of Equations. John Wiley & Sons, Inc., Feb. 1962, pp. 42-52.